

Pensieve Header: Computing the maximal order of a permutation in S_n .

```
In[1]:= $RecursionLimit = 10 000;
MaxOrder[n_] := MaxOrder[n, 2];
MaxOrder[n_, p_] /; p > n := 1;
MaxOrder[n_, p_] := MaxOrder[n, p] = Max[Table[
    p^k * MaxOrder[n - (p^k /. 1 -> 0), NextPrime[p]],
    {k, 0, Floor[Log[p, n]]}
]]


In[5]:= Print[# -> {m = MaxOrder[#], FactorInteger[m]}] & /@ Range[20];

1 -> {1, {{1, 1}}}
2 -> {2, {{2, 1}}}
3 -> {3, {{3, 1}}}
4 -> {4, {{2, 2}}}
5 -> {6, {{2, 1}, {3, 1}}}
6 -> {6, {{2, 1}, {3, 1}}}
7 -> {12, {{2, 2}, {3, 1}}}
8 -> {15, {{3, 1}, {5, 1}}}
9 -> {20, {{2, 2}, {5, 1}}}
10 -> {30, {{2, 1}, {3, 1}, {5, 1}}}
11 -> {30, {{2, 1}, {3, 1}, {5, 1}}}
12 -> {60, {{2, 2}, {3, 1}, {5, 1}}}
13 -> {60, {{2, 2}, {3, 1}, {5, 1}}}
14 -> {84, {{2, 2}, {3, 1}, {7, 1}}}
15 -> {105, {{3, 1}, {5, 1}, {7, 1}}}
16 -> {140, {{2, 2}, {5, 1}, {7, 1}}}
17 -> {210, {{2, 1}, {3, 1}, {5, 1}, {7, 1}}}
18 -> {210, {{2, 1}, {3, 1}, {5, 1}, {7, 1}}}
19 -> {420, {{2, 2}, {3, 1}, {5, 1}, {7, 1}}}
20 -> {420, {{2, 2}, {3, 1}, {5, 1}, {7, 1}}}


In[6]:= Timing[{m = MaxOrder[1000], FactorInteger[m]}]

Out[6]= {23.104, {1 516 385 558 956 728 808 659 224 171 023 365 600, {{2, 5}, {3, 3}, {5, 2}, {7, 2}, {11, 1},
    {13, 1}, {17, 1}, {19, 1}, {23, 1}, {29, 1}, {31, 1}, {37, 1}, {41, 1}, {43, 1},
    {47, 1}, {53, 1}, {59, 1}, {61, 1}, {67, 1}, {71, 1}, {73, 1}, {83, 1}, {89, 1}}}}
```

```
In[7]:= ListPlot[Log[10, MaxOrder[#]] & /@ Range[1000]]
```



```
In[8]:= ListPlot[(Log[MaxOrder[#]] / (Sqrt[#] Log[#])) & /@ Range[2, 1000]]
```



```
In[9]:= Log[2] / 2 // N
```

Out[9]= 0.346574

```
In[10]:= n = 1000; Log[MaxOrder[n]] / (Sqrt[n] Log[n]) // N
```

Out[10]= 0.381379